

Otimização de Portfólio

Jairo Macedo

Variáveis

- x_i : Valor alocado no ativo i
- v : Retorno esperado total do portfólio (Função Objetivo)

Parâmetros

- r_i : Retorno esperado do ativo i
- B : Restrição orçamentária
- R_{\max} : Limite de risco. Pode ser entendido como grau de aversão a risco do indivíduo
- d_{ij} : Elemento da matriz de variância-covariância que representa a co-variância entre os ativos i e j

Modelo

Função Objetivo

Maximizar o retorno esperado do portfólio:

$$v = \sum_i r_i x_i$$

Sujeito às Restrições

1. **Restrição Orçamentária:**

$$\sum_i x_i = B$$

2. **Restrição de Risco:**

$$\sum_i \sum_j x_i d_{ij} x_j \leq R_{\max}$$

$$x_1^2 \cdot d_{11} + x_2^2 \cdot d_{22} + x_3^2 \cdot d_{33} + 2(x_1 \cdot x_2 \cdot d_{12} + x_1 \cdot x_3 \cdot d_{13} + x_2 \cdot x_3 \cdot d_{23}) \leq R_{\max}$$

Código

```
Sets
    i /Ibov, CDI, Bitcoin/
;
alias (i,j)
Parameters
    r(i) Retorno esperado de cada ativo
        /Ibov 0.02, CDI 0.01, Bitcoin 0.03/
    B Restrição orçamentária
        /100/
    R_max Limite máximo de risco (Variância)
        /20/ ;
Table d(i,j) Matriz Var-Cov
        CDI          Ibov          Bitcoin
    CDI    0.001      -0.00005      -0.0062
    Ibov   -0.00005      0.05         0.005
    Bitcoin -0.0062      0.005         0.20 ;

Variables
    x(i) Valor alocado em cada ativo i
    v      Função objetivo a ser maximizada;

Positive Variable x;

Equations
    Objective
    BudgetConstraint
    RiskConstraint;

Objective..
        v =e= sum(i, r(i)*x(i));

BudgetConstraint..
        sum(i, x(i)) =e= B;

RiskConstraint..
        sum((i,j), x(i)*d(i,j)*x(j)) =l= R_max;

Model PortfolioOptimization /all/;

Solve PortfolioOptimization using nlp maximizing v;
```

```
Display x.l, v.l;
```

Solução

- **IBOV:** 11,872
- **CDI:** 79,77490
- **Bitcoin:** 8,353
- **Objetivo:** 1,286
- **Restrição de Risco:** 20
- **Restrição Orçamentária:** 100